Title

xtmixed postestimation — Postestimation tools for xtmixed

Description

The following postestimation commands are of special interest after `xtmixed`:

Command	Description
`estat group`	summarize the composition of the nested groups
`estat recovariance`	display the estimated random-effects covariance matrix (or matrices)
`estat icc`	estimate intraclass correlations

For information about these commands, see below.

The following standard postestimation commands are also available:

Command	Description
`contrast`	contrasts and ANOVA-style joint tests of estimates
`estat`	AIC, BIC, VCE, and estimation sample summary
`estimates`	cataloging estimation results
`lincom`	point estimates, standard errors, testing, and inference for linear combinations of coefficients
`lrtest`	likelihood-ratio test
`margins`	marginal means, predictive margins, marginal effects, and average marginal effects
`marginsplot`	graph the results from margins (profile plots, interaction plots, etc.)
`nlcom`	point estimates, standard errors, testing, and inference for nonlinear combinations of coefficients
`predict`	predictions, residuals, influence statistics, and other diagnostic measures
`predictnl`	point estimates, standard errors, testing, and inference for generalized predictions
`pwcompare`	pairwise comparisons of estimates
`test`	Wald tests of simple and composite linear hypotheses
`testnl`	Wald tests of nonlinear hypotheses

See the corresponding entries in the *Base Reference Manual* for details.

Special-interest postestimation commands

`estat group` reports number of groups and minimum, average, and maximum group sizes for each level of the model. Model levels are identified by the corresponding group variable in the data. Because groups are treated as nested, the information in this summary may differ from what you would get if you `tabulated` each group variable individually.

`estat recovariance` displays the estimated variance–covariance matrix of the random effects for each level in the model. Random effects can be either random intercepts, in which case the corresponding rows and columns of the matrix are labeled as `_cons`, or random coefficients, in which case the label is the name of the associated variable in the data.

`estat icc` displays the intraclass correlation for pairs of responses at each nested level of the model. Intraclass correlations are available for random-intercept models or for random-coefficient models conditional on random-effects covariates being equal to zero. They are not available for crossed-effects models or with residual error structures other than independent structures.

Syntax for predict

Syntax for obtaining best linear unbiased predictions (BLUPs) of random effects, or the BLUPs' standard errors

`predict` [*type*] { *stub** | *newvarlist* } [*if*] [*in*] , { `reffects` | `reses` }
  [`level(`*levelvar*`)`]

Syntax for obtaining scores after ML estimation

`predict` [*type*] { *stub** | *newvarlist* } [*if*] [*in*] , `scores`

Syntax for obtaining other predictions

`predict` [*type*] *newvar* [*if*] [*in*] [, *statistic* `level(`*levelvar*`)`]

statistic	Description
Main	
`xb`	linear prediction for the *fixed* portion of the model only; the default
`stdp`	standard error of the fixed-portion linear prediction
`fitted`	fitted values, fixed-portion linear prediction plus contributions based on predicted random effects
`residuals`	residuals, response minus fitted values
* `rstandard`	standardized residuals

Unstarred statistics are available both in and out of sample; type `predict ... if e(sample) ...` if wanted only for the estimation sample. Starred statistics are calculated only for the estimation sample, even when `if e(sample)` is not specified.

Menu

Statistics > Postestimation > Predictions, residuals, etc.

Options for predict

Main

`xb`, the default, calculates the linear prediction $\mathbf{x}\boldsymbol{\beta}$ based on the estimated fixed effects (coefficients) in the model. This is equivalent to fixing all random effects in the model to their theoretical mean value of zero.

`stdp` calculates the standard error of the linear predictor $\mathbf{x}\boldsymbol{\beta}$.

`level(`*levelvar*`)` specifies the level in the model at which predictions involving random effects are to be obtained; see the options below for the specifics. *levelvar* is the name of the model level and is either the name of the variable describing the grouping at that level or `_all`, a special designation for a group comprising all the estimation data.

`reffects` calculates best linear unbiased predictions (BLUPs) of the random effects. By default, BLUPs for all random effects in the model are calculated. However, if the `level(`*levelvar*`)` option is specified, then BLUPs for only level *levelvar* in the model are calculated. For example, if `classes` are nested within `schools`, then typing

```
. predict b*, reffects level(school)
```

would produce BLUPs at the school level. You must specify q new variables, where q is the number of random-effects terms in the model (or level). However, it is much easier to just specify *stub*`*` and let Stata name the variables *stub*`1` ... *stubq* for you.

`reses` calculates the standard errors of the best linear unbiased predictions (BLUPs) of the random effects. By default, standard errors for all BLUPs in the model are calculated. However, if the `level(`*levelvar*`)` option is specified, then standard errors for only level *levelvar* in the model are calculated; see the `reffects` option. You must specify q new variables, where q is the number of random-effects terms in the model (or level). However, it is much easier to just specify *stub*`*` and let Stata name the variables *stub*`1` ... *stubq* for you.

The `reffects` and `reses` options often generate multiple new variables at once. When this occurs, the random effects (or standard errors) contained in the generated variables correspond to the order in which the variance components are listed in the output of `xtmixed`. Still, examining the variable labels of the generated variables (using the `describe` command, for instance) can be useful in deciphering which variables correspond to which terms in the model.

`scores` calculates the parameter-level scores, one for each parameter in the model including regression coefficients and variance components. The score for a parameter is the first derivative of the log likelihood (or log pseudolikelihood) with respect to that parameter. One score per highest-level group is calculated, and it is placed on the last record within that group. Scores are calculated in the estimation metric as stored in `e(b)`.

`scores` is not available after restricted maximum-likelihood (REML) estimation.

`fitted` calculates fitted values, which are equal to the fixed-portion linear predictor *plus* contributions based on predicted random effects, or in mixed-model notation, $\mathbf{x}\boldsymbol{\beta} + \mathbf{Z}\mathbf{u}$. By default, the fitted values take into account random effects from all levels in the model; however, if the `level(`*levelvar*`)` option is specified, the fitted values are fit beginning with the topmost level down to and including level *levelvar*. For example, if `classes` are nested within `schools`, then typing

```
. predict yhat_school, fitted level(school)
```

would produce school-level predictions. That is, the predictions would incorporate school-specific random effects but not those for each class nested within each school.

`residuals` calculates residuals, equal to the responses minus fitted values. By default, the fitted values take into account random effects from all levels in the model; however, if the `level(`*levelvar*`)` option is specified, the fitted values are fit beginning at the topmost level down to and including level *levelvar*.

`rstandard` calculates standardized residuals, equal to the residuals multiplied by the inverse square root of the estimated error covariance matrix.

Syntax for estat group

```
estat group
```

Menu

Statistics > Postestimation > Reports and statistics

Syntax for estat recovariance

estat recovariance [, level(*levelvar*) correlation *matlist_options*]

Menu

Statistics > Postestimation > Reports and statistics

Options for estat recovariance

level(*levelvar*) specifies the level in the model for which the random-effects covariance matrix is to be displayed and returned in r(cov). By default, the covariance matrices for all levels in the model are displayed. *levelvar* is the name of the model level and is either the name of variable describing the grouping at that level or _all, a special designation for a group comprising all the estimation data.

correlation displays the covariance matrix as a correlation matrix and returns the correlation matrix in r(corr).

matlist_options are style and formatting options that control how the matrix (or matrices) are displayed; see [P] **matlist** for a list of what is available.

Syntax for estat icc

estat icc [, level(*#*)]

Menu

Statistics > Postestimation > Reports and statistics

Option for estat icc

level(*#*) specifies the confidence level, as a percentage, for confidence intervals. The default is level(95) or as set by set level; see [U] **20.7 Specifying the width of confidence intervals**.

Remarks

Various predictions, statistics, and diagnostic measures are available after fitting a mixed model using `xtmixed`. For the most part, calculation centers around obtaining best linear unbiased predictors (BLUPs) of the random effects. Random effects are not estimated when the model is fit but instead need to be predicted after estimation. Calculation of intraclass correlations, estimating the dependence between responses for different levels of nesting, may also be of interest.

▷ Example 1

In example 3 of [XT] **xtmixed**, we modeled the weights of 48 pigs measured on nine successive weeks as

$$\texttt{weight}_{ij} = \beta_0 + \beta_1 \texttt{week}_{ij} + u_{0j} + u_{1j}\texttt{week}_{ij} + \epsilon_{ij} \tag{1}$$

for $i = 1, \ldots, 9$, $j = 1, \ldots, 48$, $\epsilon_{ij} \sim N(0, \sigma^2_\epsilon)$, and u_{0j} and u_{1j} normally distributed with mean zero and variance–covariance matrix

$$\boldsymbol{\Sigma} = \mathrm{Var}\begin{bmatrix} u_{0j} \\ u_{1j} \end{bmatrix} = \begin{bmatrix} \sigma^2_{u0} & \sigma_{01} \\ \sigma_{01} & \sigma^2_{u1} \end{bmatrix}$$

```
. use http://www.stata-press.com/data/r12/pig
(Longitudinal analysis of pig weights)

. xtmixed weight week || id: week, covariance(unstructured) variance
  (output omitted )

Mixed-effects ML regression                     Number of obs      =       432
Group variable: id                              Number of groups   =        48

                                                Obs per group: min =         9
                                                               avg =       9.0
                                                               max =         9

                                                Wald chi2(1)       =   4649.17
Log likelihood = -868.96185                     Prob > chi2        =    0.0000

------------------------------------------------------------------------------
      weight |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
        week |   6.209896   .0910745    68.18   0.000     6.031393    6.388399
       _cons |   19.35561   .3996387    48.43   0.000     18.57234    20.13889
------------------------------------------------------------------------------

------------------------------------------------------------------------------
  Random-effects Parameters  |   Estimate   Std. Err.     [95% Conf. Interval]
-----------------------------+------------------------------------------------
id: Unstructured             |
                   var(week) |   .3715251   .0812958      .2419532     .570486
                  var(_cons) |   6.823363   1.566194      4.351297    10.69986
            cov(week,_cons)  |  -.0984378   .2545767     -.5973991    .4005234
-----------------------------+------------------------------------------------
               var(Residual) |   1.596829    .123198      1.372735    1.857505
------------------------------------------------------------------------------
LR test vs. linear regression:       chi2(3) =   764.58   Prob > chi2 = 0.0000
Note: LR test is conservative and provided only for reference.
```

Rather than see the estimated variance components listed as above, we can instead see them in matrix form; that is, we can see $\widehat{\boldsymbol{\Sigma}}$

```
. estat recovariance
Random-effects covariance matrix for level id

             |      week      _cons
-------------+----------------------
        week |  .3715251
       _cons | -.0984378   6.823363
```

or we can see $\widehat{\Sigma}$ as a correlation matrix

```
. estat recovariance, correlation
Random-effects correlation matrix for level id

             |      week      _cons
-------------+----------------------
        week |         1
       _cons | -.0618257          1
```

We can also obtain BLUPs of the pig-level random effects (u_{0j} and u_{1j}). We need to specify the variables to be created in the order `u1 u0` because that is the order in which the corresponding variance components are listed in the output (`week _cons`). We obtain the predictions and list them for the first 10 pigs.

```
. predict u1 u0, reffects
. by id, sort: generate tolist = (_n==1)
. list id u0 u1 if id <=10 & tolist

     +-------------------------------+
     | id          u0            u1  |
     |-------------------------------|
  1. |  1    .2369444    -.3957636   |
 10. |  2   -1.584127      .510038   |
 19. |  3   -3.526551     .3200372   |
 28. |  4    1.964378    -.7719702   |
 37. |  5    1.299236    -.9241479   |
     |-------------------------------|
 46. |  6   -1.147302    -.5448151   |
 55. |  7   -2.590529     .0394454   |
 64. |  8   -1.137067    -.1696566   |
 73. |  9   -3.189545    -.7365507   |
 82. | 10    1.160324     .0030772   |
     +-------------------------------+
```

If you forget how to order your variables in `predict`, or if you use `predict` *stub*`*`, remember that `predict` labels the generated variables for you to avoid confusion.

```
. describe u0 u1

              storage  display     value
variable name   type   format      label      variable label
-------------------------------------------------------------------------------
u0              float  %9.0g                  BLUP r.e. for id: _cons
u1              float  %9.0g                  BLUP r.e. for id: week
```

Examining (1), we see that, within each pig, the successive weight measurements are modeled as simple linear regression with intercept $\beta_0 + u_{j0}$ and slope $\beta_1 + u_{j1}$. We can generate estimates of the pig-level intercepts and slopes with

```
. generate intercept = _b[_cons] + u0
. generate slope = _b[week] + u1
. list id intercept slope if id<=10 & tolist
```

	id	interc~t	slope
1.	1	19.59256	5.814132
10.	2	17.77149	6.719934
19.	3	15.82906	6.529933
28.	4	21.31999	5.437926
37.	5	20.65485	5.285748
46.	6	18.20831	5.665081
55.	7	16.76509	6.249341
64.	8	18.21855	6.040239
73.	9	16.16607	5.473345
82.	10	20.51594	6.212973

Thus we can plot estimated regression lines for each of the pigs. Equivalently, we can just plot the fitted values because they are based on both the fixed and random effects:

```
. predict fitweight, fitted
. twoway connected fitweight week if id<=10, connect(L)
```



We can also generate standardized residuals and see if they follow a standard normal distribution, as they should in any good-fitting model:

```
. predict rs, rstandard
. summarize rs
    Variable |       Obs        Mean    Std. Dev.       Min        Max
-------------+--------------------------------------------------------
          rs |       432    1.01e-09    .8929356  -3.621446   3.000929
. qnorm rs
```



◁

▷ Example 2

Following Rabe-Hesketh and Skrondal (2012, chap. 2), we fit a two-level random-effects model for human peak-expiratory-flow rate. The subjects were each measured twice using the Mini-Wright peak-flow meter. It is of interest to determine how reliable the meter is as a measurement device. The intraclass correlation provides a measure of reliability. Formally, in a two-level random-effects model, the intraclass correlation corresponds to the correlation of measurements within the same individual and also to the proportion of variance explained by the individual random effect.

First, we fit the two-level model using `xtmixed`:

```
. use http://www.stata-press.com/data/r12/pefrate, clear
(Peak-expiratory-flow rate)

. xtmixed wm || id:

Performing EM optimization:

Performing gradient-based optimization:

Iteration 0:   log likelihood = -184.57839
Iteration 1:   log likelihood = -184.57839

Computing standard errors:

Mixed-effects ML regression                     Number of obs      =        34
Group variable: id                              Number of groups   =        17

                                                Obs per group: min =         2
                                                               avg =       2.0
                                                               max =         2

                                                Wald chi2(0)       =         .
Log likelihood = -184.57839                     Prob > chi2        =         .

------------------------------------------------------------------------------
          wm |      Coef.   Std. Err.      z    P>|z|     [95% Conf. Interval]
-------------+----------------------------------------------------------------
       _cons |   453.9118   26.18617    17.33   0.000     402.5878    505.2357
------------------------------------------------------------------------------

------------------------------------------------------------------------------
  Random-effects Parameters  |   Estimate   Std. Err.     [95% Conf. Interval]
-----------------------------+------------------------------------------------
id: Identity                 |
                   sd(_cons) |   107.0464   18.67858      76.04062     150.695
-----------------------------+------------------------------------------------
                sd(Residual) |   19.91083   3.414678      14.22687    27.86564
------------------------------------------------------------------------------
LR test vs. linear regression: chibar2(01) =    46.27 Prob >= chibar2 = 0.0000
```

Now we use `estat icc` to estimate the intraclass correlation:

```
. estat icc

Intraclass correlation

------------------------------------------------------------------------------
                       Level |        ICC   Std. Err.     [95% Conf. Interval]
-----------------------------+------------------------------------------------
                          id |   .9665602   .0159495      .9165853    .9870185
------------------------------------------------------------------------------
```

This correlation is close to 1, indicating that the Mini-Wright peak-flow meter is reliable. But as noted by Rabe-Hesketh and Skrondal (2012), the reliability is not just a characteristic of the instrument, but also of the between-subject variance. Here we see that the between-subject standard deviation, `sd(_cons)`, is much larger than the within-subject standard deviation, `sd(Residual)`.

In the presence of fixed-effects covariates, `estat icc` reports the residual intraclass correlation, the correlation between measurements conditional on the fixed-effects covariates. This is equivalent to the correlation of the model residuals.

In the presence of random-effects covariates, the intraclass correlation is no longer constant and depends on the values of the random-effects covariates. In this case, `estat icc` reports conditional intraclass correlations assuming zero values for all random-effects covariates. For example, in a two-level model, this conditional correlation represents the correlation of the residuals for two measurements on the same subject, which both have random-effects covariates equal to zero. Similarly to the interpretation of intercept variances in random-coefficient models (Rabe-Hesketh and Skrondal 2012, chap. 4), interpretation of this conditional intraclass correlation relies on the usefulness of the zero baseline values of random-effects covariates. For example, mean centering of the covariates is often used to make a zero value a useful reference.

◁

▷ Example 3

In example 4 of [XT] **xtmixed**, we estimated a Cobb–Douglas production function with random intercepts at the region level and at the state-within-region level:

$$\mathbf{y}_{jk} = \mathbf{X}_{jk}\boldsymbol{\beta} + u_k^{(3)} + u_{jk}^{(2)} + \boldsymbol{\epsilon}_{jk}$$

```
. use http://www.stata-press.com/data/r12/productivity
(Public Capital Productivity)
. xtmixed gsp private emp hwy water other unemp || region: || state:
  (output omitted)
```

We can use `estat group` to see how the data are broken down by state and region

```
. estat group
```

Group Variable	No. of Groups	Observations per Group Minimum	Average	Maximum
region	9	51	90.7	136
state	48	17	17.0	17

and we are reminded that we have balanced productivity data for 17 years for each state.

We can use `predict, fitted` to get the fitted values

$$\widehat{\mathbf{y}}_{jk} = \mathbf{X}_{jk}\widehat{\boldsymbol{\beta}} + \widehat{u}_k^{(3)} + \widehat{u}_{jk}^{(2)}$$

but if we instead want fitted values at the region level, that is,

$$\widehat{\mathbf{y}}_{jk} = \mathbf{X}_{jk}\widehat{\boldsymbol{\beta}} + \widehat{u}_k^{(3)}$$

we need to use the `level()` option;

```
. predict gsp_region, fitted level(region)
. list gsp gsp_region in 1/10
```

	gsp	gsp_re~n
1.	10.25478	10.40529
2.	10.2879	10.42336
3.	10.35147	10.47343
4.	10.41721	10.52648
5.	10.42671	10.54947
6.	10.4224	10.53537
7.	10.4847	10.60781
8.	10.53111	10.64727
9.	10.59573	10.70503
10.	10.62082	10.72794

❑ Technical note

Out-of-sample predictions are permitted after `xtmixed`, but if these predictions involve BLUPs of random effects, the integrity of the estimation data must be preserved. If the estimation data have changed since the mixed model was fit, `predict` will be unable to obtain predicted random effects that are appropriate for the fitted model and will give an error. Thus, to obtain out-of-sample predictions that contain random-effects terms, be sure that the data for these predictions are in observations that augment the estimation data.

❑

We can use `estat icc` to estimate residual intraclass correlations between productivity years in the same region and in the same state and region.

```
. estat icc
Residual intraclass correlation
```

Level	ICC	Std. Err.	[95% Conf.	Interval]
region	.159893	.127627	.0287143	.5506202
state|region	.8516265	.0301733	.7823466	.9016272

`estat icc` reports two intraclass correlations for this three-level nested model. The first is the level-3 intraclass correlation at the region level, the correlation between productivity years in the same region. The second is the level-2 intraclass correlation at the state-within-region level, the correlation between productivity years in the same state and region.

Conditional on the fixed-effects covariates, we find that annual productivity is only slightly correlated within the same region, but it is highly correlated within the same state and region. We estimate that state and region random effects compose approximately 85% of the total residual variance.

◁

Saved results

`estat recovariance` saves the last-displayed random-effects covariance matrix in `r(cov)` or in `r(corr)` if it is displayed as a correlation matrix.

`estat icc` saves the following in `r()`:

Scalars

`r(icc#)`	level-# intraclass correlation
`r(se#)`	standard errors of level-# intraclass correlation
`r(level)`	confidence level of confidence intervals

Macros

`r(label#)`	label for level #

Matrices

`r(ci#)`	vector of confidence intervals (lower and upper) for level-# intraclass correlation

For a G-level nested model, # can be any integer between 2 and G.

Methods and formulas

Methods and formulas are presented under the following headings:

Prediction
Intraclass correlations

Prediction

Following the notation defined throughout [XT] **xtmixed**, best linear unbiased predictions (BLUPs) of random effects $\mathbf{u}$ are obtained as

$$\widetilde{\mathbf{u}} = \widetilde{\mathbf{G}}\mathbf{Z}'\widetilde{\mathbf{V}}^{-1}\left(\mathbf{y} - \mathbf{X}\widehat{\boldsymbol{\beta}}\right)$$

where $\widetilde{\mathbf{G}}$ and $\widetilde{\mathbf{V}}$ are $\mathbf{G}$ and $\mathbf{V} = \mathbf{Z}\mathbf{G}\mathbf{Z}' + \sigma^2_\epsilon\mathbf{R}$ with ML or REML estimates of the variance components plugged in. Standard errors for BLUPs are calculated based on the iterative technique of Bates and Pinheiro (1998, sec. 3.3) for estimating the BLUPs themselves. If estimation is done by REML, these standard errors account for uncertainty in the estimate of $\boldsymbol{\beta}$, while for ML the standard errors treat $\boldsymbol{\beta}$ as known. As such, standard errors of REML-based BLUPs will usually be larger.

Fitted values are given by $\mathbf{X}\widehat{\boldsymbol{\beta}} + \mathbf{Z}\widetilde{\mathbf{u}}$, residuals as $\widehat{\boldsymbol{\epsilon}} = \mathbf{y} - \mathbf{X}\widehat{\boldsymbol{\beta}} - \mathbf{Z}\widetilde{\mathbf{u}}$, and standardized residuals as

$$\widehat{\boldsymbol{\epsilon}}_* = \widehat{\sigma}_\epsilon^{-1}\widehat{\mathbf{R}}^{-1/2}\widehat{\boldsymbol{\epsilon}}$$

If the `level(`*levelvar*`)` option is specified, fitted values, residuals, and standardized residuals consider only those random-effects terms up to and including level *levelvar* in the model.

For details concerning the calculation of scores, see *Methods and formulas* in [XT] **xtmixed**.

Intraclass correlations

Consider a simple two-level random-intercept model:

$$y_{ij} = \beta + u_j^{(2)} + \epsilon_{ij}^{(1)}$$

for measurements $i = 1, \dots, n_j$ and level-2 groups $j = 1, \dots, M$, where y_{ij} is a response, β is an unknown fixed intercept, u_j is a level-2 random intercept, and $\epsilon_{ij}^{(1)}$ is a level-1 error term. Errors are assumed to be normally distributed with mean zero and variance σ_1^2; random intercepts are assumed to be normally distributed with mean zero and variance σ_2^2 and independent of error terms.

The intraclass correlation for this model is

$$\rho = \mathrm{Corr}(y_{ij}, y_{i'j}) = \frac{\sigma_2^2}{\sigma_1^2 + \sigma_2^2}$$

It corresponds to the correlation between measurements i and i' from the same group j.

Now consider a three-level nested random-intercept model:

$$y_{ijk} = \beta + u_{jk}^{(2)} + u_k^{(3)} + \epsilon_{ijk}^{(1)}$$

for measurements $i = 1, \ldots, n_{jk}$ and level-2 groups $j = 1, \ldots, M_{1k}$ nested within level-3 groups $k = 1, \ldots, M_2$. Here $u_{jk}^{(2)}$ is a level-2 random intercept, $u_k^{(3)}$ is a level-3 random intercept, and $\epsilon_{ijk}^{(1)}$ is a level-1 error term. The error terms and random intercepts are assumed to be normally distributed with mean zero and variances σ_1^2, σ_2^2, and σ_3^2, respectively, and to be mutually independent.

We can consider two types of intraclass correlations for this model. We will refer to them as level-2 and level-3 intraclass correlations. The level-3 intraclass correlation is

$$\rho^{(3)} = \text{Corr}(y_{ijk}, y_{i'j'k}) = \frac{\sigma_3^2}{\sigma_1^2 + \sigma_2^2 + \sigma_3^2}$$

This is the correlation between measurements i and i' from the same level-3 group k and from different level-2 groups j and j'.

The level-2 intraclass correlation is

$$\rho^{(2)} = \text{Corr}(y_{ijk}, y_{i'jk}) = \frac{\sigma_2^2 + \sigma_3^2}{\sigma_1^2 + \sigma_2^2 + \sigma_3^2}$$

This is the correlation between measurements i and i' from the same level-3 group k and level-2 group j. (Note that level-1 intraclass correlation is undefined.)

More generally, for a G-level nested random-intercept model, the g-level intraclass correlation is defined as

$$\rho^{(g)} = \frac{\sum_{l=g}^{G} \sigma_l^2}{\sum_{l=1}^{G} \sigma_l^2}$$

The above formulas also apply in the presence of fixed-effects covariates $\mathbf{X}$ in a random-effects model. In this case, intraclass correlations are conditional on fixed-effects covariates and are referred to as residual intraclass correlations. `estat icc` also uses the same formulas to compute intraclass correlations for random-coefficient models, assuming zero baseline values for the random-effects covariates, and labels them as conditional intraclass correlations. The above formulas assume independent residual structures.

Intraclass correlations are estimated using the delta method and will always fall in (0,1) because variance components are nonnegative. To accommodate the range of an intraclass correlation, we use the logit transformation to obtain confidence intervals.

Let $\widehat{\rho}^{(g)}$ be a point estimate of the intraclass correlation and $\widehat{\text{SE}}(\widehat{\rho}^{(g)})$ its standard error. The $(1-\alpha) \times 100\%$ confidence interval for $\text{logit}(\rho^{(g)})$ is

$$\text{logit}(\widehat{\rho}^{(g)}) \pm z_{\alpha/2} \frac{\widehat{\text{SE}}(\widehat{\rho}^{(g)})}{\widehat{\rho}^{(g)}(1-\widehat{\rho}^{(g)})}$$

where $z_{\alpha/2}$ is the $1-\alpha/2$ quantile of the standard normal distribution and $\mathrm{logit}(x) = \ln\{x/(1-x)\}$. Let k_u be the upper endpoint of this interval, and let k_l be the lower. The $(1-\alpha)\times 100\%$ confidence interval for $\rho^{(g)}$ is then given by

$$\left(\frac{1}{1+e^{-k_l}}, \frac{1}{1+e^{-k_u}}\right)$$

References

Bates, D. M., and J. C. Pinheiro. 1998. Computational methods for multilevel modelling. In *Technical Memorandum BL0112140-980226-01TM*. Murray Hill, NJ: Bell Labs, Lucent Technologies. http://stat.bell-labs.com/NLME/CompMulti.pdf.

Rabe-Hesketh, S., and A. Skrondal. 2012. *Multilevel and Longitudinal Modeling Using Stata*. 3rd ed. College Station, TX: Stata Press.

Also see

[XT] **xtmixed** — Multilevel mixed-effects linear regression

[U] **20 Estimation and postestimation commands**